

GenRef  
v1.00

MDOS Reference guide.

TIPI Library

(C) Copyright 2023  
Beery W. Miller  
**ALL RIGHTS RESERVED**

---

**TIPI - CONTENTS**


---

# Contents

<b>TIPI OVERVIEW.....</b>	<b>4</b>
<b>CALLING TIPI FUNCTIONS.....</b>	<b>5</b>
<b>RESET TIPI.....</b>	<b>6</b>
<b>IDENTIFY CRU OF THE TIPI .....</b>	<b>7</b>
<b>OPEN HOSTNAME .....</b>	<b>8</b>
<b>RECEIVE TIPI MESSAGE .....</b>	<b>9</b>
<b>SEND TIPI MESSAGE .....</b>	<b>10</b>
<b>CLOSE TIPI CONNECTION .....</b>	<b>11</b>
<b>RECEIVE TIPI STRING .....</b>	<b>12</b>
<b>SEND ANSI STRING (UP ARROW KEY) .....</b>	<b>13</b>
<b>SEND ANSI STRING (DOWN ARROW KEY) .....</b>	<b>14</b>
<b>SEND ANSI STRING (LEFT ARROW KEY).....</b>	<b>15</b>
<b>SEND ANSI STRING (RIGHT ARROW KEY) .....</b>	<b>16</b>
<b>SEND ANSI STRING (PAGEUP KEY).....</b>	<b>17</b>
<b>SEND ANSI STRING (PAGEDOWN KEY) .....</b>	<b>18</b>
<b>SERVER BIND PORT .....</b>	<b>19</b>
<b>SERVER UNBIND .....</b>	<b>20</b>
<b>SERVER ACCEPT.....</b>	<b>21</b>
<b>TIPI MOUSE MOVEMENT .....</b>	<b>22</b>
<b>SEND TCP STRING .....</b>	<b>23</b>
<b>SEND TCP STRING .....</b>	<b>24</b>
<b>OPEN UDP URL .....</b>	<b>25</b>
<b>CLOSE UDP CONNECTION .....</b>	<b>26</b>
<b>SEND UDP MESSAGE.....</b>	<b>27</b>
<b>RECEIVE UDP MESSAGE .....</b>	<b>28</b>
<b>OPEN TLS URL .....</b>	<b>29</b>

**CLOSE TLP CONNECTION ..... 30**  
**SEND TLP MESSAGE..... 31**  
**RECEIVE TLS MESSAGE ..... 32**  
**TIPI LOG EXTENSION ..... 33**

---

**TIPI Overview**

---

All TIPI management routines in MDOS are provided to aid a programmer in writing applications requiring TIPI operations beyond the immediate instruction set of the TMS 9995 microprocessor. The following TIPI operations are currently supported within the operating system.

Type	Extension	Description
0x20	Mouse	Read USB mouse data
0x22	TCP	Client/Server socket connections
0x23	UDP	Client UDP socket connections
0x24	TLS	Client TLS socket connections
0x25	LOG	Add message to /var/log/tipi/tipi.log

Opcodes within the TIPI XOP Library 12 support access to the USB mouse, as a wired or wireless connection. In addition, both client and server based TCP, UDP, and TLS connections are supported. The TIPI is a continually evolving hardware and software solution for the TI-99/4A and Geneve 9640 computers.

For more information on the TIPI, please visit <https://github.com/jedimatt42/tipi/wiki>

---

## CALLING TIPI FUNCTIONS

---

The MDOS TIPI Library must be called from within a machine code program running as a task under MDOS. You pass arguments to the TIPI Library via the calling registers.

The MDOS TIPI Library is invoked from a machine code program when software trap number zero (XOP 0) is called with a library number of 12. The calling program's R0 must contain the 16-bit subprogram at the time of the XOP. The following code fragment will open a TCP connection to a hostname or IP address.

```

      LI      R0,>0002    Open hostname
      LI      R1,HOST     Hostname to open with port number
      LI      R2,22       Length of Hostname String
      LI      R3,>0100    TCP Socket Handle Byte (MSB)
      XOP     @TIPI,0     Access subprogram

      TIPI    DATA      12
      HOST    TEXT      "Heatwave.ddns.net:9640"    HOST to open
      EVEN

```

In the preceding example, two hidden assumptions were made. First it is assumed that Hostname or IP address is located on a page which is currently mapped into a memory page which has the same 16-bit address page number as its Virtual address page number (read the section on Memory Management). The second assumption is the TIPI label is currently mapped into a memory page which has the same 16-bit address page number as its Virtual address page number.

## TIPI Library

### Reset TIPI

**Function** Sends a command to the TIPI to kill the TipiService on the Raspberry PI and restarts it asynchronously. When this command is issued, allow time for the TipiService to resume.

**Parameters** R0 = >0000 (opcode)

**Results** N/A -

#### Parameter Description

Not applicable

#### Example Code

```
LI      R0,>0000    Reset TIPI opcode
XOP     @TIPI,0     Access subprogram
```

TIPI DATA 12 TIPI Library

<b>Identify CRU of the TIPI</b>
---------------------------------

**Function** Returns the CRU address of the TIPI.

**Parameters** R0 = >0001

**Results** R0 = Result code

**Parameter Description**

Result code CRU address of the TIPI  
 CRU = >0000, no TIPI present

**Example Code**

LI	R0,>0001	CRU address opcode
XOP	@TIPI,0	Access subprogram
MOV	R0,@CRU	Store CRU Address of TIPI
TIPI	DATA 12	TIPI Library
CRU	DATA 0	CRU of TIPI

<b>Open Hostname</b>
----------------------

**Function**                      Opens a TCP socket to a Hostname or IP address at a specified port.

**Parameters**                      R0 = >0002  
                                         R1 = Hostname or IP address  
                                         R2 = Length of R1  
                                         R3 = Socket handle (MSB)

**Results**                          R1 = Result code

**Parameter Description**

Hostname or IP	The hostname or IP address is a string in the form of "hostname:port" without the quotes.
Length	The length of the string Hostname:Port
Socket handle	Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.
Result code	>FFFF if connect  >0000 for no connection

**Example Code**

	LI	R0,>0002		Open Host opcode					
	LI	R1,HOST1		Host or IP address to open with port number					
	LI	R2,22		Length of Host String					
	LI	R3,>0100		TCP Socket Handle Byte (MSB)					
	XOP	@TIPI,0		Access subprogram					
	MOV	R0,R0		Test for connection					
	JEQ	NOCONCT		No connection,					
TIPI	DATA	12		TIPI Library					
HOST1	TEXT	"Heatwave.ddns.net:9640"		Host to open					
	EVEN								
HOST2	TEXT	'localhost:23'		Host to open					
	EVEN								
HOST3	TEXT	'192.168.1.79:9640'		Host to open					
	EVEN								



<b>Receive TCP Message</b>
----------------------------

**Function**                      Receive a single character from the TIPI TCP Socket Messaging system

**Parameters**                      R0 = >0003  
                                         R1 = Socket handle (MSB)

**Results**                              R0 = Result code  
                                         R1 = Received character (MSB)

**Parameter Description**

Socket handle                      Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

Result code                              >0000 no character received  
     >FFFF if character received

Received character                      Received character from >00 to >FF received from the socket handle

**Example Code**

```

LOOP
    LI      R0,>0003      Receive a single character
    LI      R1,>0100      TCP Socket Handle Byte (MSB)
    XOP     @TIPI,0       Access subprogram

    MOV     R0,R0         Test for status
    JEQ     LOOP          No character, run through loop again
    MOVB    R1,CHAR       Move character to buffer

TIPI  DATA 12           TIPI Library
CHAR  DATA 0           Character to receive

```

<b>Send TCP Message</b>
-------------------------

**Function**                      Send a single character from the TIPI TCP Socket Messaging system.

**Parameters**                      R0 = >0004  
                                         R1 = Character to send (MSB)  
                                         R2 = Socket handle (MSB)

**Results**                          Not applicable

**Parameter Description**

Character to Send                  A single byte from >00 to >FF to send via the TIPI messaging system.

Socket handle                      Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LI      R0,>0004      Send single character
MOV     @CHAR,R1      Character to send
LI      R2,>0100      Socket handle
XOP     @TIPI,0        Access subprogram

```

```

TIPI    DATA 12       TIPI Library
CHAR    DATA >0D00    Character to send (MSB)

```

<b>Close TIPI Connection</b>
------------------------------

**Function**                      Close an open TIPI TCP Socket.

**Parameters**                  R0 = >0005  
                                      R1 = Socket handle (MSB)

**Results**                        N/A

**Parameter Description**

Socket handle                  Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LOOP
    LI      R0,>0005      Close TIPI TCP Socket
    LI      R1,>0100      TCP Socket Handle Byte (MSB)
    XOP     @TIPI,0       Access subprogram

TIPI  DATA 12           TIPI Library
  
```

<b>Receive TIPI String</b>
----------------------------

**Function**                      Receive multiple characters from the TIPI TCP Socket Messaging system.

**Parameters**                      R0 = >0006  
                                         R1 = String Pointer  
                                         R2 = String length  
                                         R3 = Socket handle (MSB)

**Results**                          R0 = Result code

**Parameter Description**

String Pointer                      An address in memory where data will be returned to a buffer in the program's physical 64K memory space.

String length                        The length of the buffer in memory to be processed. The Raspberry PI maintains a larger buffer passing up to the maximum amount of characters in the program's buffer. Upon return from the TIPI XOP call, the actual number of characters retrieved will be passed.

The string length shall not exceed the boundary of a 2<sup>nd</sup> page mapped in memory from the first physical page as only 2 x 8K pages are mapped into use during the TIPI XOP opcode call.

Socket handle                        Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

Result code                          >0000 no character  
                                         >0001 or greater, characters returned to the buffer.

**Example Code**

```

LOOP
    LI    R0,>0006    Receive string
    LI    R1,BUF      Buffer containing the characters
    LI    R2,8192      Length of buffer
    LI    R3,>0100     TCP Socket Handle Byte (MSB)
    XOP   @TIPI,0      Access subprogram

    MOV   R0,R0        Test for status
    JEQ   LOOP         No character(s) received, run through loop again

TIPI    DATA 12       TIPI Library
BUF     BSS   8192     Buffer
  
```

<b>Send ANSI String (Up Arrow Key)</b>
--

**Function**                      Send an ANSI sequence for the Down Arrow key.

**Parameters**                      R0 = >0007  
                                         R1 = Socket handle (MSB)

**Results**                          Not applicable

**Parameter Description**

Socket handle                      Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LOOP
    LI      R0,>0007      Send ANSI Sequence for Up-Arrow keypress
    LI      R1,>0100      TCP Socket Handle Byte (MSB)
    XOP     @TIPI,0       Access subprogram

TIPI  DATA 12           TIPI Library
  
```

<b>Send ANSI String (Down Arrow Key)</b>
--

**Function**                      Send an ANSI sequence for the Down Arrow key.

**Parameters**                      R0 = >0008  
                                         R1 = Socket handle (MSB)

**Results**                          Not applicable

**Parameter Description**

Socket handle                      Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LOOP
    LI      R0,>0008      Send ANSI Sequence for Down-Arrow keypress
    LI      R1,>0100      TCP Socket Handle Byte (MSB)
    XOP     @TIPI,0       Access subprogram

TIPI  DATA 12           TIPI Library
  
```

<b>Send ANSI String (Left Arrow Key)</b>
--

**Function**                      Send an ANSI sequence for the Left Arrow key.

**Parameters**                      R0 = >0009  
                                         R1 = Socket handle (MSB)

**Results**                          Not applicable

**Parameter Description**

Socket handle                      Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LOOP
    LI      R0,>0009      Send ANSI Sequence for Left-arrow keypress
    LI      R1,>0100      TCP Socket Handle Byte (MSB)
    XOP     @TIPI,0       Access subprogram

TIPI  DATA 12           TIPI Library

```

<b>Send ANSI String (Right Arrow Key)</b>
---

**Function**                      Send an ANSI sequence for the Right Arrow key.

**Parameters**                      R0 = >000A  
                                         R1 = Socket handle (MSB)

**Results**                          Not applicable

**Parameter Description**

Socket handle                      Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LOOP
    LI      R0,>000A    Send ANSI Sequence for Right-Arrow keypress
    LI      R1,>0100    TCP Socket Handle Byte (MSB)
    XOP     @TIPI,0     Access subprogram

TIPI  DATA 12         TIPI Library
  
```



<b>Send ANSI String (PageUp Key)</b>
--------------------------------------

**Function**                      Send an ANSI sequence for the PageUp Arrow key.

**Parameters**                      R0 = >000B  
                                         R1 = Socket handle (MSB)

**Results**                          Not applicable

**Parameter Description**

Socket handle                      Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LOOP
    LI      R0,>000B      Send ANSI Sequence for PageUp-Arrow keypress
    LI      R1,>0100      TCP Socket Handle Byte (MSB)
    XOP     @TIPI,0       Access subprogram

TIPI  DATA 12           TIPI Library
  
```

<b>Send ANSI String (PageDown Key)</b>
--

**Function**                      Send an ANSI sequence for the Up Arrow key.

**Parameters**                      R0 = >000C  
                                         R1 = Socket handle (MSB)

**Results**                          Not applicable

**Parameter Description**

Socket handle                      Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LOOP
    LI      R0,>000C    Send ANSI Sequence for Up-Arrow keypress
    LI      R1,>0100    TCP Socket Handle Byte (MSB)
    XOP     @TIPI,0     Access subprogram

TIPI  DATA 12         TIPI Library
  
```

<b>Server Bind Port</b>
-------------------------

**Function**                      Bind a listening server socket to a listening port.

**Parameters**                      R0 = >000D  
                                         R1 = Interface address (IP:Port) string  
                                         R2 = Interface address String length  
                                         R3 = Server handle (MSB)

**Results**                          R0 = Result code

**Parameter Description**

Interface Port String              An address in memory containing the Interface Port to be binded. An interface port can be specified by the address such as “\*:9640” excluding quotes to bind port 9640 on the Raspberry PI.

Interface length                    The length of the interface port in memory to be binded. The Raspberry PI maintains a larger buffer passing up to the maximum amount of characters in the program's buffer. Upon return from the TIPI XOP call, the actual number of characters retrieved will be passed.

The string length shall not exceed the boundary of a 2<sup>nd</sup> page mapped in memory from the first physical page as only 2 x 8K pages are mapped into use during the TIPI XOP opcode call.

Server handle                      Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple server sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state. This handle should not be confused with the client socket handle.

Result code                          >0000 binding not successful  
                                         >FFFF binding successful

**Example Code**

```

LI      R0,>000D    Bind Server Port
LI      R1,SERV     Buffer containing the characters for the interface port
MOV     @LEN,R2     Length of the interface port string to be binded
LI      R3,>0100    TCP Socket Handle Byte (MSB)
XOP     @TIPI,0     Access subprogram

MOV     R0,R0       Test for status
JNE     FOUND       Server binded, let's go
BLWP    @0          Server not binded, let's exit

```

```

TIPI    DATA 12     TIPI Library
LEN     DATA 6
SERVR   TEXT "*:9640"

```

<b>Server Unbind</b>
----------------------

**Function**                      Unbind a previously binded server.

**Parameters**                  R0 = >000E  
                                      R1 = Server handle (MSB)

**Results**                        Not applicable

**Parameter Description**

Socket handle                  Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LOOP
    LI      R0,>000E      Unbind server
    LI      R1,>0100      TCP Socket Handle Byte (MSB)
    XOP     @TIPI,0       Access subprogram

TIPI  DATA 12           TIPI Library
  
```

<b>Server Accept</b>
----------------------

**Function**                      Accept an incoming connection on a previously binded server interface address.

**Parameters**                  R0 = >000F  
                                      R1 = Server handle (MSB)

**Results**                        R0 = Result code (MSB)

**Parameter Description**

Server handle                  Server handles will have a server handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple server handles can be managed by the TIPI XOP within a single program or with several programs in a multitasking state. This server handle should not be confused with the socket handle.

Result code                    >0000 no incoming socket  
                                      >FF00 error on server handle  
                                      >0100 to >FE00, Socket handle of the MSB

Socket handle                  Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LOOP
    LI      R0,>000F      Accept incoming connection
    LI      R1,>0100      Server handle
    XOP     @TIPI,0       Access subprogram

    MOV     R0,R0         Test for status
    JEQ     LOOP         No connection, loop again as still waiting
    MOV     R0,SOCK       Store socket handle

TIPI  DATA 12           TIPI Library
SOCK  DATA 0           Socket handle

```

<b>TIPI Mouse Movement</b>
----------------------------

**Function**                      Get Mouse release data. This data call mimics the Video XOP opcode call >32.

**Parameters**                  R0x = >0010

**Results**                      R1x = Returned X displacement  
                                     R2x = Returned Y displacement  
                                     R3x = Button State

**Parameter Description**

Returned X displacement      Returns the x direction displacement from the last call.

Returned Y displacement      Returns the y direction displacement from the last call.

Button State                    The button state will return the left, middle, and right button status.

b1 b2 b3 0 xxxx xxxx xxxx (MSB bits)

b1 = left                      1 = down

b2 = middle                  1 = down

b3 = right                    1 = down

**Example Code**

	LI	R0,>0010	Send single character
	XOP	@TIPI,0	Access subprogram
	MOV	R1,@XDIS	X distance movement
	MOV	R2,@YDIS	Y Distance movement
	MOV	R3,@BUT	Mouse button status
TIPI	DATA	12	TIPI Library
XDIS	DATA	0	X Distance movement
YDIS	DATA	0	Y Distance movement
BUT	DATA	0	Button status

<b>Send TCP String</b>
------------------------

**Function**                      Send TIPI TCP message, maximum 10 characters. Useful when sending a response sequence when a single byte one at a time has timing issues with the receiving system.

**Parameters**                      R0 = >0011  
                                         R1 = String length contained in R2,R3,R4,R5,R6 (max 10 bytes)  
                                         R2 = Chars  
                                         R3 = Chars  
                                         R4 = Chars  
                                         R5 = Chars  
                                         R6 = Chars  
                                         R7 = Socket handle (MSB)

**Results**                          R0 = >0000 no character, greater than or equal to >0001 or more characters returned to the buffer.

**Parameter Description**

String length                      The length of the buffer in the registers in the current workspace in registers R2 to R6 to be processed. The maximum length of the buffer is 10 characters.

Socket handle                      Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LI      R0,>0011    Send String
LI      R1,8        Length of string to send
*      Build string containing "Client#1"
LI      R2,'Cl'      Two characters, "Cl"
LI      R3,'ie'      Two characters, "ie"
LI      R4,'nt'      Two characters, "nt"
LI      R5,'#1'      Two characters, "#1"
CLR     R6          Clear register, but not required to be cleared
LI      R7,>0100     TCP Socket Handle Byte (MSB)
XOP     @TIPI,0     Access subprogram

TIPI    DATA  12    TIPI XOP Library

```

<b>Send TCP String</b>
------------------------

**Function**                      Send TCP TCP message, maximum 8K characters. Useful when sending a response sequence when a single byte one at a time has timing issues with the receiving system.

**Parameters**                      R0 = >0012  
                                         R1 = String pointer to write  
                                         R2 = String length beginning after first three bytes of the string pointer destination.

**Results**                          R0 = >0000 no character, greater than or equal to >0001 or more characters returned to the buffer.

**Parameter Description**

String length                      String length is defined as:

1 <sup>st</sup> Byte	>22
2 <sup>nd</sup> Byte	Socket handle byte >01 to >FF
3 <sup>rd</sup> Byte	>03 (write command)
4 <sup>th</sup> Byte to 8K	String up to 8K

Socket handle                      Socket handles will have a socket handle of a single byte from >01 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LI      R0,>0013    Send String
LI      R1,HELLO    String Pointer
LI      R2,7        String length
XOP     @TIPI,0     Access subprogram

```

```

HELLO BYTE >22      Write
          BYTE >00    Socket handle
          BYTE >03    Write
          TEXT 'Hello'
          BYTE >0D,>0A  cr/lf

```

```

TIPI    DATA 12    TIPI XOP Library

```



<b>Open UDP URL</b>
---------------------

**Function**                      Opens a TIPI UDP socket to a Hostname or IP address at a specified port.

**Parameters**                      R0 = >0013  
    R1 = Hostname or IP address  
    R2 = Length of R1  
    R3 = Socket handle (MSB)

**Results**                              R1 = Result code

**Parameter Description**

Hostname or IP	The hostname or IP address is a string in the form of "hostname:port" without the quotes.
Length	The length of the string Hostname:Port
Socket handle	Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.
Result code	>FFFF if connect  >0000 for no connection

**Example Code**

	LI	R0,>0013	Open Host opcode	
	LI	R1,HOST1	Host or IP address to open with port number	
	LI	R2,22	Length of Host String	
	LI	R3,>0100	TCP Socket Handle Byte (MSB)	
	XOP	@TIPI,0	Access subprogram	
	MOV	R0,R0	Test for connection	
	JEQ	NOCONCT	No connection,	
TIPI	DATA	12	TIPI Library	
HOST1	TEXT	"Heatwave.ddns.net:9640"	Host to open	
	EVEN			
HOST2	TEXT	'localhost:23'	Host to open	
	EVEN			
HOST3	TEXT	'192.168.1.79:9640'	Host to open	
	EVEN			

<b>Close UDP Connection</b>
-----------------------------

**Function**                      Close an open TIPI UDP Socket.

**Parameters**                  R0 = >0014 (opcode)  
                                      R1 = Socket handle (MSB)

**Results**                        N/A

**Parameter Description**

Socket handle                  Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LOOP
    LI      R0,>0014      Close TIPI TCP Socket
    LI      R1,>0100      TCP Socket Handle Byte (MSB)
    XOP     @TIPI,0       Access subprogram

TIPI  DATA 12           TIPI Library
  
```

<b>Send UDP Message</b>
-------------------------

**Function**                      Send TIPI UDP message, maximum 8K characters.

**Parameters**                      R0 = >0015  
                                         R1 = String pointer to write  
                                         R2 = String length beginning after first three bytes of the string pointer destination.

**Results**                              R0 = >0000 all characters sent, not >0000 then error on sending string.

**Parameter Description**

String length                      String length is defined as:

1 <sup>st</sup> Byte	>23
2 <sup>nd</sup> Byte	Socket handle byte >01 to >FF
3 <sup>rd</sup> Byte	>03 (write command)
4 <sup>th</sup> Byte to 8K	String up to 8K

Socket handle                      Socket handles will have a socket handle of a single byte from >01 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LI      R0,>0015    Send String
LI      R1,HELLO    String Pointer
LI      R2,7        String length
XOP     @TIPI,0     Access subprogram

HELLO BYTE >23      Write
        BYTE >00     Socket handle
        BYTE >03     Write
        TEXT 'Hello'
        BYTE >0D,>0A  cr/lf

```

TIPI    DATA   12        TIPI XOP Library

<b>Receive UDP Message</b>
----------------------------

**Function**                      Receive a single character from the TIPI UDP Socket Messaging system

**Parameters**                      R0 = >0016  
                                         R1 = String Pointer  
                                         R2 = String length  
                                         R3 = Socket handle (MSB)

**Results**                          R0 = Result code

**Parameter Description**

String Pointer                      An address in memory where data will be returned to a buffer in the program's physical 64K memory space.

String length                      The length of the buffer in memory to be processed. The Raspberry PI maintains a larger buffer passing up to the maximum amount of characters in the program's buffer. Upon return from the TIPI XOP call, the actual number of characters retrieved will be passed.

The string length shall not exceed the boundary of a 2<sup>nd</sup> page mapped in memory from the first physical page as only 2 x 8K pages are mapped into use during the TIPI XOP opcode call.

Socket handle                      Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

Result code                          >0000 no character  
                                         >0001 or greater, characters returned to the buffer.

**Example Code**

```

LOOP
    LI    R0,>0016    Receive string
    LI    R1,BUF      Buffer containing the characters
    LI    R2,8192      Length of buffer
    LI    R3,>0100     TCP Socket Handle Byte (MSB)
    XOP   @TIPI,0      Access subprogram

    MOV   R0,R0        Test for status
    JEQ   LOOP         No character(s) received, run through loop again

TIPI    DATA 12       TIPI Library
BUF     BSS   8192     Buffer
  
```

<b>Open TLS URL</b>
---------------------

**Function**                      Opens a TIPI TLS socket to a Hostname or IP address at a specified port.

**Parameters**                      R0 = >0017  
                                         R1 = Hostname or IP address  
                                         R2 = Length of R1  
                                         R3 = Socket handle (MSB)

**Results**                          R1 = Result code

**Parameter Description**

Hostname or IP	The hostname or IP address is a string in the form of "hostname:port" without the quotes.
Length	The length of the string Hostname:Port
Socket handle	Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.
Result code	>FFFF if connect  >0000 for no connection

**Example Code**

```

LI      R0,>0017    Open Host opcode
LI      R1,HOST1    Host or IP address to open with port number
LI      R2,22        Length of Host String
LI      R3,>0100     TCP Socket Handle Byte (MSB)
XOP     @TIPI,0      Access subprogram
MOV     R0,R0        Test for connection
JEQ     NOCONCT      No connection,

TIPI    DATA 12      TIPI Library

HOST1 TEXT "Heatwave.ddns.net:9640"    Host to open
      EVEN
HOST2 TEXT 'localhost:23'              Host to open
      EVEN
HOST3 TEXT '192.168.1.79:9640'          Host to open
      EVEN

```

<b>Close TLP Connection</b>
-----------------------------

**Function**                      Close an open TIPI TLP Socket.

**Parameters**                  R0 = >0018 (opcode)  
                                      R1 = Socket handle (MSB)

**Results**                      N/A

**Parameter Description**

Socket handle                  Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LOOP
    LI      R0,>0018      Close TIPI TCP Socket
    LI      R1,>0100      TCP Socket Handle Byte (MSB)
    XOP     @TIPI,0       Access subprogram

TIPI  DATA 12           TIPI Library
  
```

<b>Send TLP Message</b>
-------------------------

**Function**                      Send a TIPI TLP message, maximum 8K characters.

**Parameters**                      R0 = >0019  
                                         R1 = String pointer to write  
                                         R2 = String length beginning after first three bytes of the string pointer destination.

**Results**                              R0 = >0000 all characters sent, not >0000 then error on sending string.

**Parameter Description**

String length                      String length is defined as:

1 <sup>st</sup> Byte	>24
2 <sup>nd</sup> Byte	Socket handle byte >01 to >FF
3 <sup>rd</sup> Byte	>03 (write command)
4 <sup>th</sup> Byte to 8K	String up to 8K

Socket handle                      Socket handles will have a socket handle of a single byte from >01 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

**Example Code**

```

LI      R0,>0019    Send String
LI      R1,HELLO    String Pointer
LI      R2,7        String length
XOP     @TIPI,0     Access subprogram

```

```

HELLO BYTE >24     TLP Extension
        BYTE >00    Socket handle
        BYTE >03    Write
        TEXT 'Hello'
        BYTE >0D,>0A  cr/lf

```

TIPI    DATA   12        TIPI XOP Library

<b>Receive TLS Message</b>
----------------------------

**Function**                      Receive a single character from the TIPI TLS Socket Messaging system

**Parameters**                      R0 = >001A  
    R1 = String Pointer  
    R2 = String length  
    R3 = Socket handle (MSB)

**Results**                          R0 = Result code

**Parameter Description**

String Pointer                      An address in memory where data will be returned to a buffer in the program's physical 64K memory space.

String length                        The length of the buffer in memory to be processed. The Raspberry PI maintains a larger buffer passing up to the maximum amount of characters in the program's buffer. Upon return from the TIPI XOP call, the actual number of characters retrieved will be passed.

The string length shall not exceed the boundary of a 2<sup>nd</sup> page mapped in memory from the first physical page as only 2 x 8K pages are mapped into use during the TIPI XOP opcode call.

Socket handle                        Socket handles will have a socket handle of a single byte from >00 to >FF, assigned arbitrarily by the code it passes in the open command. Multiple sockets can be managed by the TIPI XOP within a single program or with several programs in a multitasking state.

Result code                          >0000 no character  
    >0001 or greater, characters returned to the buffer.

**Example Code**

```

LOOP
    LI    R0,>001A    Receive string
    LI    R1,BUF      Buffer containing the characters
    LI    R2,8192      Length of buffer
    LI    R3,>0100     TCP Socket Handle Byte (MSB)
    XOP   @TIPI,0      Access subprogram

    MOV   R0,R0        Test for status
    JEQ   LOOP         No character(s) received, run through loop again

TIPI    DATA 12       TIPI Library
BUF     BSS   8192     Buffer

```



<b>TIPI Log Extension</b>
---------------------------

**Function**                      Writes a message to the “tipi.log” file on the TIPI. Useful for Debugging purposes.

**Parameters**                      R0 = >001B  
                                         R1 = String Pointer  
                                         R2 = String length

**Results**                          Not Applicable

**Parameter Description**

String length                      String length is defined as:

1 <sup>st</sup> Byte	>25
2 <sup>nd</sup> Byte to 8K	String up to 8K

**Example Code**

```

LOOP
  LI      R0,>0025    Receive a single character
  LI      R1,HELLO    String Pointer
  LI      R2,7        String length
  XOP     @TIPI,0      Access subprogram

  HELLO BYTE >25      TLP Extension
           TEXT 'Hello'
           BYTE >0D,>0A  cr/lf

  TIPI    DATA 12     TIPI XOP Library

```